DOCUMENT RESUME

ED 415 243 TM 027 898

AUTHOR	Blankmeyer, Eric
TITLE	Robust Estimates of Location and Dispersion.
PUB DATE	1997-00-00
NOTE	13p.
PUB TYPE	Reports - Descriptive (141)
EDRS PRICE	MF01/PC01 Plus Postage.
DESCRIPTORS	*Estimation (Mathematics); *Regression (Statistics); *Robustness (Statistics); Statistical Analysis
IDENTIFIERS	*Dispersion (Statistics); *Locational Analysis; Median (Statistics)

ABSTRACT

This paper gives concise descriptions of a robust location statistic, the remedian of P. Rousseeuw and G. Bassett (1990) and a robust measure of dispersion, the "Sn" of P. Rousseeuw and C. Croux (1993). The use of Sn in least absolute errors regression (L1) is discussed, and BASIC programs for both statistics are provided. The remedian is an iterated median that needs a small amount of memory to process large data sets. It is an attractive robust estimator of location in large samples where the storage of the individual observations is impractical. The Sn estimator is unbiased and has an asymptotic efficiency of about 58%. Among its many applications is the question of a scale estimate for L1 regression problems, those that minimize the sum of the absolute residuals instead of the sum of the squared residuals. Appendixes contain both BASIC programs. (Author/SLD)

Robust Estimates of Location and Dispersion

Eric Blankmeyer
Department of Finance and Economics
Southwest Texas State University
San Marcos, TX 78666
e-mail: eb01@business.swt.edu
telephone 512-245-3253

Abstract. This paper gives concise descriptions of a robust location statistic, Rousseeuw and Bassett's remedian, and a robust measure of dispersion, Rousseeuw and Croux's Sn. The use of Sn in least absolute errors regression (L1) is discussed, and Basic programs for both statistics are provided.

Key words. ANOVA, L1 regression, median, robust statistics

Robust Estimates of Location and Dispersion

Introduction

During the last four of five decades, statisticians have devoted a great deal of attention to robust estimation. Indeed, Tukey has described the "almost parametric models" as a third wave of statistics, following upon parametric and non-parametric theory (Koenker and Bassett 1978, p. 35). Advances in computer hardware and algorithms now allow researchers to apply these statistics to large data sets and to use procedures that combine robustness with high efficiency.

Researchers need to be aware of these developments because classical statistical methods are quite vulnerable to outliers in data. The average and the standard deviation can be broken down by just a few anomalous observations. In other words, our usual estimates of location and dispersion may have large biases if the outliers remain undetected. This possibility arises especially in large data sets which are processed entirely by computer without a personal inspection by the researcher.

Aberrant observations may result from unusual events, measurement errors, faulty transcription, or simply from a noisy error distribution. The outliers themselves may be of considerable interest if they represent meaningful exceptions --for example, an academic program which produces outstanding test scores. Because the average is not a robust measure of location, outliers may not show up in the residuals scattered around the average. This situation may be aggravated by a masking effect, in which one anomalous observation hides several other outliers. It follows that scrutiny of the residuals may not be effective unless robust estimates of location and dispersion have been computed.

The next section of this paper describes the remedian, a robust location measure for large data sets. Then a robust dispersion measure, Sn, is presented; and its role in L1 regression is discussed. The appendices contain Basic programs for these statistics. The programs are based on algorithms developed by Gilbert Bassett, Jr., Cristophe Croux, Annick M. Leroy, and Peter J. Rousseeuw.

Location: the remedian

Robust statistics use more computer time and memory than their nonrobust least-squares counterparts, a fact which has impeded robust estimation for very large data sets. A sample average, for example, requires little storage since the individual observations need not be retained in memory; each is just added to the running sum and then discarded. Computation of a median, on the other hand, requires that all the observations be available in memory for sorting.

Rousseeuw and Bassett (1990) proposed the remedian, an iterated median which needs a small amount of memory to process large data sets. In the Basic version (Appendix A), 17 observations are read at a time, and their median is computed and stored in a cell. The observations are then discarded, and the next 17 data are read. These steps are repeated until the entire sample has been processed. Using just five arrays, each with 17 observations, the program can process as many as 17^5 = 1,419,857 numbers.

How well does the remedian deal with outliers ? One important indicator of robustness is the breakdown point, the amount of contamination an estimator can withstand. In a sample of n observations, just one outlier, if it is bad enough, can carry the average and the standard deviation indefinitely far from their population values, so the breakdown point of those classical statistics is only 1/n. The sample median's bias, on the other hand, remains finite when as many as half the observations undergo arbitrary contamination. Therefore, the median's breakdown point approaches fifty percent. This is the maximum breakdown point for an affine-equivariant estimator; if more than half the sample is corrupted, the estimator cannot tell the good observations from the bad ones.

Rousseeuw and Bassett find that the remedian's breakdown point can be low when the outliers happen to clump together in one or a few groups. If, on the other hand, the outliers are located randomly throughout the sample, the probability of breakdown is quite small. Furthermore, the authors show that the remedian estimates the population median consistently and is equivariant under monotonic transformations of the data. The remedian is therefore an attractive robust estimator of location in large samples, where storage of the individual observations is impractical.

Dispersion: the Sn statistic

Because it is based on least squares, the standard deviation has a low breakdown point (asymptotically zero); a few stray observations can make it explode. Alternatively, one can compute the average absolute residual around the median, which also fails to be a robust measure of dispersion. The median of the absolute residuals does achieve the maximum breakdown point and is fairly easy to compute, but its efficiency is low (only about 37 percent). Rousseeuw and Croux (1993) proposed an high-breakdown alternative to the standard deviation. Denoted Sn, the estimator is unbiased and has an asymptotic efficiency of about 58 percent. For n observations x(i), Sn is simply proportional to the double median over all n(n-1)/2 differences | x(i) - x(j) | . Accordingly, Sn resembles a mode or "nearest neighbor" statistic. As such, it resists skewness and outliers. For moderate or large n, the computations would be quite burdensome were it not for Rousseeuw and Croux's clever algorithm.

Among the many applications of Sn is the question of a scale estimate for L1 regression problems, those which minimize the sum of the absolute residuals instead of the sum of the squared residuals (Dodge 1987, Dodge 1992). The L1 regression has low breakdown if there are bad leverage points (extreme observations) among the independent variables. In that case, the least median of squares criterion or one of its variants is preferable (Rousseeuw and Leroy 1987). In other situations, however, bad leverage points are precluded, and the outliers are confined to the dependent variable. This is true for polynomial trends and for many designed experiments, including ANOVA models. In these cases, the L1 regression attains the maximum breakdown point. For large samples and numerous independent variables, the L1 regression is also easier to compute than the least median of squares since the former can be solved as an ordinary linear program or by special algorithms (Dodge 1992).

What role does the Sn statistic play here ? Bassett and Koenker (1978) showed that the large-sample covariance matrix for L1 regression differs from least squares only in the dispersion parameter. A principal obstacle to robust inference for the L1 regression has therefore been the choice of a reasonably efficient, high-breakdown alternative to the standard deviation. This issue has been examined by McKean and Schrader (1987) and Koenker (1987), among other authors.

It is well known that the L1 regression always produces several zero residuals, as many as there are free parameters in the model. The zero residuals represent the observations "used up" in the estimation process. It seems natural to discard those zeros and then apply Sn to the nonzero residuals. The resulting "standard error of the estimate" would have large-sample validity when used in the usual regression and ANOVA hypothesis tests.

Appendix B contains a Basic program to compute Sn. The online resource STATLIB has Rousseeuw and Croux's FORTRAN version of Sn.

References

Bassett, Jr., Gilbert and Roger Koenker (1978). "Asymptotic Theory of Least Absolute Error Regression," *Journal of the American Statistical Association*, Vol. 73, pp. 618-622.

Dodge, Yadolah (editor) (1987). *Statistical Data Analysis Based on the L1-Norm and Related Methods*. Amsterdam, Netherlands: North-Holland.

Dodge, Yadolah (editor) (1992). L1-Statistical Analysis and Related Methods. Amsterdam, Netherlands: North-Holland.

Koenker, Roger (1987). "A Comparison of Asymptotic Testing Methods for L1-Regression" in Dodge (1987), pp. 287-295.

Koenker, Roger and Gilbert Bassett, Jr. (1978). "Regression Quantiles," *Econometrica*, Vol. 46, No. 1, pp. 33-50.

McKean, Joseph W. and Ronald M. Schrader (1987). "Least Absolute Errors Analysis of Variance" in Dodge (1987), pp. 297-305.

Rousseeuw, Peter J. and Annick M. Leroy (1987). *Robust Regression and Outlier Detection*. New York: Wiley.

Rousseeuw, Peter J. and Gilbert W. Bassett, Jr. (1990). "The Remedian: A Robust Averaging Method for Large Data Sets," *Journal of the American Statistical Association*, Vol. 85, No. 409, pp. 97-104.

Rousseeuw, Peter J. and Christophe Croux (1993). "Alternatives to the Median Absolute Deviation," *Journal of the American Statistical Association*, Vol. 88, No. 424, pp. 1273-1283.

Appendix A. A Basic Program for the Remedian

```
REM  THIS PROGRAM COMPUTES THE REMEDIAN, A MEDIAN FOR
REM  LARGE DATA SETS. THE REMEDIAN, WHICH USES A
REM  SMALL AMOUNT OF COMPUTER STORAGE, WAS PROPOSED BY:
REM    PETER J. ROUSSEEUW AND GILBERT W. BASSETT, JR.
REM    (1990), "THE REMEDIAN: A ROBUST AVERAGING
REM    METHOD FOR LARGE DATA SETS," JOURNAL OF THE
REM    AMERICAN STATISTICAL ASSOCIATION, 85, 97-104.
REM
REM  THE CURRENT VERSION OF THIS PROGRAM CAN HANDLE
REM  AS MANY AS 17^5 = 1,419,857 OBSERVATIONS.
REM  THE FUNCTION FMED, WHICH FINDS AN ORDER STATISTIC,
REM  WAS CREATED BY DR. ANNICK LEROY AND APPEARS IN
REM  "PROGRESS," A FORTRAN PROGRAM FOR ROBUST REGRESSION.
REM  THE CODE FOR SORTING A VECTOR OF DATA (LINES 600-
REM  834) WAS ALSO ADAPTED FROM THE "PROGRESS" PROGRAM.
REM
REM  THIS PROGRAM CARRIES NO WARRANTY, EXPRESSED OR
REM  IMPLIED. THE PROGRAM'S SUITABILITY FOR COMMERCIAL
REM  USE OR FOR ANY PARTICULAR APPLICATION IS NOT
REM  GUARANTEED.
REM
080 DEFLNG I-N
090 DECLARE FUNCTION FMED(Y())
100 PRINT "WHAT IS THE INPUT FILE ?"
110 INPUT INFILE$
120 PRINT "WHAT IS THE OUTPUT FILE ?"
130 INPUT OUTFILE$
140 PRINT "HOW MANY OBSERVATIONS (<= 1,419,857) ?"
150 INPUT NOBS
160 OPEN INFILE$ FOR INPUT AS #1
170 OPEN OUTFILE$ FOR OUTPUT AS #2
180 DIM A1(17),A2(17),A3(17),A4(17),A5(17)
182 DIM W(85),V(85),JLV(85),JRV(85)
190 I = 0
192 FOR N = 1 TO 17
200 FOR M = 1 TO 17
210 FOR L = 1 TO 17
220 FOR K = 1 TO 17
230 FOR J = 1 TO 17
240 I = I + 1
242 IF I > NOBS GOTO 340
250 INPUT #1, ADATUM
260 A1(J) = ADATUM
270 NEXT J
280 A2(K) = FMED(A1())
290 NEXT K
300 A3(L) = FMED(A2())
310 NEXT L
320 A4(M) = FMED(A3())
330 NEXT M
332 A5(N) = FMED(A4())
334 NEXT N
340 JW = J-1
```

```
350 KW = K-1
360 LW = L-1
370 MW = M-1
372 NW = N-1
380 II=0
390 FOR I = 1 TO JW
400 II=II+1
410 V(II) = A1(I)
420 W(II) = 1.0
430 NEXT I
440 FOR I = 1 TO KW
450 II=II+1
460 V(II) = A2(I)
470 W(II) = 17.0
480 NEXT I
490 FOR I = 1 TO LW
500 II=II+1
510 V(II) = A3(I)
520 W(II) = 289.0
530 NEXT I
540 FOR I = 1 TO MW
550 II=II+1
560 V(II) = A4(I)
570 W(II) = 4913.0
580 NEXT I
582 FOR I = 1 TO NW
584 II=II+1
586 V(II) = A5(I)
588 W(II) = 83521.0
590 NEXT I
592 NTOT = JW+KW+LW+MW+NW
600 JSS = 1
610 JLV(1) = 1
620 JRV(1) = NTOT
630 JNDL = JLV(JSS)
640 JR = JRV(JSS)
650 JSS = JSS - 1
660 JNC = JNDL
670 J = JR
680 JTWE = (JNDL + JR) / 2
690 XX = V(JTWE)
700 IF V(JNC) >= XX GOTO 730
710 JNC = JNC + 1
720 GOTO 700
730 IF XX >= V(J) GOTO 760
740 J = J - 1
750 GOTO 730
760 IF JNC > J GOTO 806
770 AMM = V(JNC)
772 WTEMP = W(JNC)
780 V(JNC) = V(J)
782 W(JNC) = W(J)
790 V(J) = AMM
792 W(J) = WTEMP
```

```
800 JNC = JNC + 1
804 J = J - 1
806 IF JNC <= J GOTO 700
808 IF (J - JNDL) < (JR - JNC) GOTO 822
810 IF JNDL >= J GOTO 818
812 JSS = JSS + 1
814 JLV(JSS) = JNDL
816 JRV(JSS) = J
818 JNDL = JNC
820 GOTO 832
822 IF JNC >= JR GOTO 830
824 JSS = JSS + 1
826 JLV(JSS) = JNC
828 JRV(JSS) = JR
830 JR = J
832 IF JNDL < JR GOTO 660
834 IF JSS <> 0 GOTO 630
850 SUMW = 0.0
852 WCAP = (NOBS+1)/2
854 FOR I = 1 TO NTOT
856 IF SUMW >= WCAP GOTO 870
858 SUMW = SUMW + W(I)
860 NEXT I
870 REMEDIAN = V(I-1)
880 PRINT #2, "REMEDIAN ="; REMEDIAN
890 END

970 FUNCTION FMED (Y())
975 NH = 9
980 LL = 1
990 LR = 17
1000 IF LL >= LR GOTO 1210
1010 AX = Y(NH)
1020 JNC = LL
1030 J = LR
1040 IF JNC > J GOTO 1180
1050 IF Y(JNC) >= AX GOTO 1080
1060 JNC = JNC + 1
1070 GOTO 1050
1080 IF Y(J) <= AX GOTO 1110
1090 J = J - 1
1100 GOTO 1080
1110 IF JNC > J GOTO 1170
1120 WA = Y(JNC)
1130 Y(JNC) = Y(J)
1140 Y(J) = WA
1150 JNC = JNC + 1
1160 J = J - 1
1170 GOTO 1040
1180 IF J < NH THEN LL = JNC
1190 IF NH < JNC THEN LR = J
1200 GOTO 1000
1210 FMED = Y(NH)
1220 END FUNCTION
```

Appendix B. A Basic Program for the Sn Statistic

```
REM  THIS PROGRAM COMPUTES THE ROBUST DISPERSION ESTIMATE,
REM  Sn. THE FUNCTIONS AND SUBROUTINES IN THIS PROGRAM
REM  ARE ADAPTED FROM THE FORTRAN CODE "SN.FOR"
REM  CREATED BY PETER J. ROUSSEEUW AND HIS
REM  COLLEAGUES. THE THEORY AND STATISTICAL PROPERTIES
REM  OF THE ROBUST SCALE ESTIMATE SN ARE DISCUSSED
REM  IN THE FOLLOWING PAPER:
REM
REM    P. J. ROUSSEEUW AND C. CROUX (1993):
REM    "ALTERNATIVES TO THE MEDIAN ABSOLUTE DEVIATION,"
REM    JOURNAL OF THE AMERICAN STATISTICAL ASSOCIATION,
REM    88, 1273-1283.
REM
REM  THERE IS NO WARRANTY, EXPRESSED OR IMPLIED, FOR THIS
REM  PROGRAM. ITS SUITABILITY FOR COMMERCIAL USE OR FOR
REM  ANY PARTICULAR PURPOSE IS NOT GUARANTEED.
REM
005 DEFLNG I-N
010 DECLARE FUNCTION SCALE(Y(),N)
015 DECLARE FUNCTION FMED(Y(),N)
020 DECLARE SUB SORT(Y(),N)
025 PRINT "WHAT IS THE INPUT FILE ?"
030 INPUT INFILE$
035 PRINT "WHAT IS THE OUTPUT FILE ?"
040 INPUT OUTFILE$
045 PRINT "HOW MANY OBSERVATIONS ?"
050 INPUT N
055 OPEN INFILE$ FOR INPUT AS #1
060 OPEN OUTFILE$ FOR OUTPUT AS #2
065 DIM X(N)
070 FOR I = 1 TO N
075 INPUT #1, X(I)
085 NEXT I
086 SN = SCALE(X(),N)
088 PRINT #2, "ROBUST SCALE SN =  ",SN
090 END


1500 FUNCTION SCALE(Y(),N)
1502 DIM A2(N)
1504 CALL SORT(Y(),N)
1506 A2(1) = Y(N\2+1)-Y(1)
1508 FOR I = 2 TO (N+1)\2
1510 NA=I-1
1512 NB=N-I
1514 IDIFF=NB-NA
1516 LEFTA=1
1518 LEFTB=1
1520 IRIGHTA=NB
1522 IRIGHTB=NB
1524 IAMIN=IDIFF\2+1
1526 IAMAX=IDIFF\2+NA
1528 REM
```

```
1530 IF LEFTA < IRIGHTA THEN
1532 LENGTH=IRIGHTA-LEFTA+1
1534 IEVEN=1-(LENGTH MOD 2)
1536 IHALF=(LENGTH-1)\2
1538 ITRYA=LEFTA+IHALF
1540 ITRYB=LEFTB+IHALF
1542 IF ITRYA < IAMIN THEN
1544 IRIGHTB=ITRYB
1546 LEFTA=ITRYA+IEVEN
1548 ELSE
1550 IF ITRYA > IAMAX THEN
1552 IRIGHTA=ITRYA
1554 LEFTB=ITRYB+IEVEN
1556 ELSE
1558 RMEDA=Y(I)-Y(I-ITRYA+IAMIN-1)
1560 RMEDB=Y(ITRYB+I)-Y(I)
1562 IF RMEDA >= RMEDB THEN
1564 IRIGHTA=ITRYA
1566 LEFTB=ITRYB+IEVEN
1568 ELSE
1570 IRIGHTB=ITRYB
1572 LEFTA=ITRYA+IEVEN
1574 END IF
1576 END IF
1578 END IF
1580 GOTO 1528
1582 END IF
1584 IF LEFTA > IAMAX THEN
1586 A2(I) = Y(LEFTB+I)-Y(I)
1588 ELSE
1600 RMEDA=Y(I)-Y(I-LEFTA+IAMIN-1)
1602 RMEDB=Y(LEFTB+I)-Y(I)
1604 A2(I) = RMEDA
1606 IF RMEDB < RMEDA THEN
1608 A2(I) = RMEDB
1610 END IF
1612 END IF
1614 NEXT I
1616 FOR I = (N+1)\2+1 TO N-1
1618 NA=N-I
1620 NB=I-1
1622 IDIFF=NB-NA
1624 LEFTA=1
1626 LEFTB=1
1628 IRIGHTA=NB
1630 IRIGHTB=NB
1632 IAMIN=IDIFF\2+1
1634 IAMAX=IDIFF\2+NA
1636 REM
1638 IF LEFTA < IRIGHTA THEN
1640 LENGTH=IRIGHTA-LEFTA+1
1642 IEVEN=1-(LENGTH MOD 2)
1644 IHALF=(LENGTH-1)\2
1646 ITRYA=LEFTA+IHALF
```

```
1648 ITRYB=LEFTB+IHALF
1650 IF ITRYA < IAMIN THEN
1652 IRIGHTB=ITRYB
1654 LEFTA=ITRYA+IEVEN
1656 ELSE
1658 IF ITRYA > IAMAX THEN
1660 IRIGHTA=ITRYA
1662 LEFTB=ITRYB+IEVEN
1664 ELSE
1666 RMEDA=Y(I+ITRYA-IAMIN+1)-Y(I)
1668 RMEDB=Y(I)-Y(I-ITRYB)
1670 IF RMEDA >= RMEDB THEN
1672 IRIGHTA=ITRYA
1674 LEFTB=ITRYB+IEVEN
1676 ELSE
1678 IRIGHTB=ITRYB
1680 LEFTA=ITRYA+IEVEN
1682 END IF
1684 END IF
1686 END IF
1688 GOTO 1636
1690 END IF
1692 IF LEFTA > IAMAX THEN
1694 A2(I)=Y(I)-Y(I-LEFTB)
1696 ELSE
1698 RMEDA=Y(I+LEFTA-IAMIN+1)-Y(I)
1700 RMEDB=Y(I)-Y(I-LEFTB)
1702 A2(I)=RMEDA
1704 IF RMEDB < RMEDA THEN
1706 A2(I)=RMEDB
1708 END IF
1710 END IF
1712 NEXT I
1714 A2(N)=Y(N)-Y((N+1)\2)
1716 CN=1
1718 IF N <= 9 THEN
1720 IF N = 2 THEN CN = 0.743
1722 IF N = 3 THEN CN = 1.851
1724 IF N = 4 THEN CN = 0.954
1726 IF N = 5 THEN CN = 1.351
1728 IF N = 6 THEN CN = 0.993
1730 IF N = 7 THEN CN = 1.198
1734 IF N = 8 THEN CN = 1.005
1736 IF N = 9 THEN CN = 1.131
1738 ELSE
1740 IF (N MOD 2) = 1 THEN CN = N/(N-0.9)
1742 END IF
1744 SCALE = CN*1.1926*FMED(A2(),N)
1746 END FUNCTION
```

```
594 SUB SORT(Y(),N)
596 DEFLNG I-N
598 DIM JLV(N), JRV(N)
600 JSS = 1
610 JLV(1) = 1
620 JRV(1) = N
630 JNDL = JLV(JSS)
640 JR = JRV(JSS)
650 JSS = JSS - 1
660 JNC = JNDL
670 J = JR
680 JTWE = (JNDL + JR) \ 2
690 XX = Y(JTWE)
700 IF Y(JNC) >= XX GOTO 730
710 JNC = JNC + 1
720 GOTO 700
730 IF XX >= Y(J) GOTO 760
740 J = J - 1
750 GOTO 730
760 IF JNC > J GOTO 806
770 AMM = Y(JNC)
780 Y(JNC) = Y(J)
790 Y(J) = AMM
800 JNC = JNC + 1
804 J = J - 1
806 IF JNC <= J GOTO 700
808 IF (J - JNDL) < (JR - JNC) GOTO 822
810 IF JNDL >= J GOTO 818
812 JSS = JSS + 1
814 JLV(JSS) = JNDL
816 JRV(JSS) = J
818 JNDL = JNC
820 GOTO 832
822 IF JNC >= JR GOTO 830
824 JSS = JSS + 1
826 JLV(JSS) = JNC
828 JRV(JSS) = JR
830 JR = J
832 IF JNDL < JR GOTO 660
834 IF JSS <> 0 GOTO 630
836 END SUB

970 FUNCTION FMED (Y(),N)
975 DEFLNG I-N
980 LL = 1
990 LR = N
995 NH = (N+1)\2
1000 IF LL >= LR GOTO 1210
1010 AX = Y(NH)
1020 JNC = LL
1030 J = LR
1040 IF JNC > J GOTO 1180
1050 IF Y(JNC) >= AX GOTO 1080
1060 JNC = JNC + 1
```

```
1070 GOTO 1050
1080 IF Y(J) <= AX GOTO 1110
1090 J = J - 1
1100 GOTO 1080
1110 IF JNC > J GOTO 1170
1120 WA = Y(JNC)
1130 Y(JNC) = Y(J)
1140 Y(J) = WA
1150 JNC = JNC + 1
1160 J = J - 1
1170 GOTO 1040
1180 IF J < NH THEN LL = JNC
1190 IF NH < JNC THEN LR = J
1200 GOTO 1000
1210 FMED = Y(NH)
1220 END FUNCTION
```

U.S. Department of Education
Office of Educational Research and Improvement (OERI)
Educational Resources Information Center (ERIC)

ERIC

TM027898

REPRODUCTION RELEASE

(Specific Document)

I. DOCUMENT IDENTIFICATION:

Title: ROBUST ESTIMATES OF LOCATION AND DISPERSION

Author(s): ERIC BLANKMEYER

Corporate Source:

Publication Date: 1997

II. REPRODUCTION RELEASE:

In order to disseminate as widely as possible timely and significant materials of interest to the educational community, documents announced in the monthly abstract journal of the ERIC system, *Resources in Education* (RIE), are usually made available to users in microfiche, reproduced paper copy, and electronic/optical media, and sold through the ERIC Document Reproduction Service (EDRS) or other ERIC vendors. Credit is given to the source of each document, and, if reproduction release is granted, one of the following notices is affixed to the document.

If permission is granted to reproduce and disseminate the identified document, please CHECK ONE of the following two options and sign at the bottom of the page.

[✓] Check here **For Level 1 Release:** Permitting reproduction in microfiche (4" x 6" film) or other ERIC archival media (e.g., electronic or optical) *and* paper copy.

The sample sticker shown below will be affixed to all Level 1 documents

PERMISSION TO REPRODUCE AND DISSEMINATE THIS MATERIAL HAS BEEN GRANTED BY

Sample

TO THE EDUCATIONAL RESOURCES INFORMATION CENTER (ERIC)

Level 1

The sample sticker shown below will be affixed to all Level 2 documents

PERMISSION TO REPRODUCE AND DISSEMINATE THIS MATERIAL IN OTHER THAN PAPER COPY HAS BEEN GRANTED BY

Sample

TO THE EDUCATIONAL RESOURCES INFORMATION CENTER (ERIC)

Level 2

[ ] Check here **For Level 2 Release:** Permitting reproduction in microfiche (4" x 6" film) or other ERIC archival media (e.g., electronic or optical), but *not* in paper copy.

Documents will be processed as indicated provided reproduction quality permits. If permission to reproduce is granted, but neither box is checked, documents will be processed at Level 1.

"I hereby grant to the Educational Resources Information Center (ERIC) nonexclusive permission to reproduce and disseminate this document as indicated above. Reproduction from the ERIC microfiche or electronic/optical media by persons other than ERIC employees and its system contractors requires permission from the copyright holder. Exception is made for non-profit reproduction by libraries and other service agencies to satisfy information needs of educators in response to discrete inquiries."

Sign here→ please

Signature: Eric Blankmeyer

Printed Name/Position/Title: ERIC BLANKMEYER PROFESSOR OF ECONOMICS

Organization/Address: DEPT. FINANCE/ECONOMICS SOUTHWEST TEXAS STATE U. SAN MARCOS, TX 78666

Telephone: 512 245 3253

FAX:

E-Mail Address: EB01@BUSINESS.SWT.EDU

Date: 11-19-97

(over)